
Running head: SUPPLEMENTARY MATERIALS 1

Supplementary materials for ”Changes in students’ self-efficacy when learning a new

topic in mathematics: A micro-longitudinal study”

Karin Street

Western Norway University of Applied Sciences, Norway

Lars-Erik Malmberg

University of Oxford, UK

Gabriel Stylianides

University of Oxford, UK

24 May 2022



SUPPLEMENTARY MATERIALS 2

Supplementary materials for ”Changes in students’ self-efficacy when learning a new

topic in mathematics: A micro-longitudinal study”

Supplementary Materials

In these supplementary materials we present (1) initial confirmatory factor

analysis (CFA) as tests of structural validity of our measurement, and (2) the steps for

specifying our multivariate multilevel models for change (MVMLMC) as reported in the

main manuscript.

Structural validity

Figure 1 . Confirmatory Factor Analysis of Self-efficacies for Easy, Medium and Hard
Tasks
Note: Ellipses represent latent factors. The rectangles represent manifest items Double-headed
arrows between manifest items represent correlated uniquenesses for facets of mathematics
The introductory phrases and items were:
”In the next lesson in maths I can learn <geometry> ...”
1 ... if I get lots of help from the teacher
2 ... if I get some help from the teacher
3 ... without any help from the teacher at all ”In the next lesson in maths I can solve ... ”
4 ... all the easy tasks
5 ... all the medium difficulty tasks
6 ... all the hard tasks
”In the next lesson in maths I can manage not to give up... ”
7 ... if I am faced with slightly difficult tasks
8 ... if I am faced with medium difficulty tasks
9 ... if I am faced with very difficult tasks

We tested a series of CFAs, as depicted in Fig 1. The nine items (see footnote in

Fig 1) were divided into three facets: (1) learning with the support of the teacher, (2)

being able to solve tasks, and (3) not giving up in the face of difficulty. We posed a

three-factor solution for level of difficulty, i.e., Easy, Medium and Hard tasks. The three

”easy” items (1,4, and 7) loaded on the ”Easy” factor. The three ”medium” items (2,5,

and 8) loaded on the ”Medium” factor. The three ”hard” items (3,6, and 9) loaded on

the ”Hard” factor. The uniquenesses of the three items were allowed to correlate.

As indices of good model fit we used Root Mean Square Error of Approximation

(RMSEA) <.06, SRMR <.08, CFI >.95 for acceptable model fit (Hu & Bentler, 1999),

and parameter estimates (small standard errors, meaningful and within-bounds

estimates (Marsh, Scalas & Nagengast, 2010). Models for each of the lessons fitted data

well.

Table 1
Fit indices for confirmatory factor analyses: Self-Efficacy lesson 1-4

Model χ2 df p RMSEA CFI TLI SRMR
1 Lesson 1 (n=162) 26.68 15 0.031 0.07 0.98 0.96 0.03
2 Lesson 2 (n=148) 25.40 15 0.045 0.07 0.98 0.96 0.03
3 Lesson 3 (n=157) 32.69 15 0.005 0.09 0.98 0.94 0.03
4 Lesson 4 (n=126) 23.83 15 0.068 0.07 0.99 0.97 0.05

Note: We modelled latent constructs for self-efficacy for easy, medium and
hard tasks, with correlated uniquenesses for facets.

Format of data

We organised the three self-efficacy variables for easy, medium and hard tasks, as

long data with nti = 596, ni = 170. There were thus four time-points nested within each

student, and each student had three dependent variables. We estimated the level

(intercept at the first lesson) and change (linear and quadratic slopes) over time.

Models

As shown in Fig 2, we specified a series of MVMLMCs. In our baseline model (M0)

we inspected the variance components for self-efficacy, by specifying additional variables

in our model (i.e., ICC = between-group variance divided by the total variance).

In the first model (M1) we estimated fixed effects of time (i.e., the first to the

fourth lesson). Time was centred at the first lesson giving linear effect codes of 0,1,2,3

and quadratic effects of 0,1,4,9. The linear effects detect increase or decrease over time,

while the quadratic effects detect accelerations or decelerations over time, visible in

curvatures in graphic depiction of the growth/change over time. At the within-level

self-efficacy for easy, medium and hard tasks were regressed on linear and quadratic

time. The variance of the disturbance terms are estimated for each DV, and these are

allowed to be associated.

At the between-level the intercepts are estimated for each self-efficacy construct

(the triangle including ”1”). The constructs are represented as circles as the variables

are unobserved, i.e., they are estimated by the model, not available in the data. This

mean/intercept is interpreted as the mean of each self-efficacy belief in the first lesson

(at the zero point in time). The variance of each construct indicates individual

differences in self-efficacy i.e., how much individual student’s starting-points varies from

the average starting-point.

 Multivariate Multilevel Models for Change

Figure 2 . Multivariate multilevel models for change
Note: Rectangles depict observed scale-scores and covariates. Time (lessons) was coded
0-1-2-3 for interpretation of the latent intercept at the beginning of each sequence of lessons.
Time was included as a linear effect (Time) and quadratic effect (Time2). Ellipses indicate
latent variables (intercepts and slopes) estimated at the between-level. Difficulty-level of
self-efficacy was indicated in subscripts for easy, medium and hard (Easy, Med, Hard).
Single-headed arrows are directional (regression) paths, and double-headed arrows either
associations, variances, or residual variances. A solid black dot on a regression-path indicates
a random effect, estimated at the between-level as a latent construct, e.g., ”Change SE”
indicating change in self-efficacy over time. Individual differences in starting-point and change
in self-efficacy over time is captured by the variances of the latent intercepts and slopes
respectively. At the between-level the triangle including ”1” represents the mean (intercept) of
the construct when the predictor is zero, and is interpreted as the level of each self-efficacy
belief in the first lesson.

Model 2

In the second model we included fixed and random effects of time. The random

effects mean that the slope of each individual student’s linear change was allowed to

differ from the overall slope. At the within-level we indicate the random effect using a

”black dot” on the regression paths from lesson to the three self-efficacy variables. The

average slope is estimated as a latent construct at the between-level. Individual

differences in slopes (i.e., how much students’ slopes differ from the average slope?) are

estimated by the variances of the latent change constructs (”Change SE”). All

intercepts and slopes are allowed to be associated at the between-level.



SUPPLEMENTARY MATERIALS 6

M3: Grade (year-group) effects

Figure 3 . Multivariate multilevel models for change
Note: Year-group 0 = grade 6, 1 = grade 10. There were three classrooms in which students
learned different content-areas: fractions (three grade 6 classes), measurement (one grade 6
classes), geometry (three grade 10 classes) and equations (three grade 10 classes). In model 4a
mathematics-content was dummy-code (0 = fractions, 1 = measurement) and in 4b (0 =
algebra, 1 = geometry).

Model 3

In the third model we included grade (year-group) at the between level as a

dichotomous variable: 0 = grade 6, 1 = grade 10, and estimated its effects on level and

change in the three self-efficacy constructs.

Model 4

In the fourth model we carried out the analyses within each year-group separately.

In the grade 6 group we defined classes in which students were learning fractions (three

classrooms) as the baseline group and included a dummy-variable for those who were

learning measurement (one classroom). In the grade 10 group we defined classes in

which students were learning algebra (three classrooms) as the baseline group and

included a dummy-variable for those who were learning geometry (two classrooms). We

included the dummy-coded variable for the content area as predictor of starting-point

and change in the three self-efficacy constructs.

In the following we present the annotated syntaxes used in MPlus8.6 (Muthén &

Muthén, 1998-2017).

TITLE: Multivariate multilevel models for change (MVMLMC),
baseline model (M0) 24 May 2022 ;

DATA: FILE = C:\se.txt ;

VARIABLE: NAMES =
id classid ! student id, classroom id,
mathtop ! content area in maths,
c1_frac6 c2_frac6 c3_frac6 c4_mea6 ! classroom dummy variables
c5_eq10 c6_eq10 c7_eq10 ! classroom dummy variables
c8_geo10 c9_geo10 ! classroom dummy variables
sex grade lesson ! student sex, grade, lesson
seE seM seH ; ! selfefficacy for easy,

! medium and hard tasks ;
USEVAR = seE seM seH ; ! variables included in model
CLUSTER = id ; ! clustering variable
MISSING ARE ALL (-9); ! code for missing value

ANALYSIS: TYPE = TWOLEVEL ; ! two-level model
ESTIMATOR = BAYES ; ! Bayesian estimator
FBITERATIONS = 10000 ; ! number of iterations
PROC = 2 ; ! number of processors
BSEED = 463956 ; ! random seed
THIN = 10 ; ! thinning

MODEL:
%WITHIN% ! within-part of the model
seE WITH seM seH ; ! associated variables
seM WITH seH ; ! associated variables
seE (wv_E) ; ! within level variances with labels
seM (wv_M) ;
seH (wv_H) ;

%BETWEEN% ! between-part of the model
seE WITH seM seH ; ! associated latent intercepts
seM WITH seH ; ! associated latent intercepts
seE (bv_E) ; ! between level variances with labels
seM (bv_M) ;
seH (bv_H) ;

MODEL CONSTRAINTS: ! additional variables (intraclass correlations, ICC)
NEW (easy_icc, med_icc, hard_icc) ; ! three ICCs
easy_icc = wv_E / (wv_E + bv_E) ; ! ICC for SE for easy tasks
med_icc = wv_M / (wv_M + bv_M) ; ! ICC for SE for medium tasks
hard_icc = wv_H / (wv_H + bv_H) ; ! ICC for SE for hard tasks

OUTPUT:
CINTERVAL (HPD) STAND TECH1 TECH4 TECH8 ;

PLOT:
TYPE = PLOT2 PLOT3;

!SAVEDATA:
!FILE IS multivar0_out.txt ; ! save outfile
!SAVE = FSCORES (1000) ; ! save factor-score based on 1000 simulations

The intraclass correlations, i.e., the proportion of variance at the between-level

was ρICC = 0.81, 95 % credibility interval (C.I.) [0.77,0.85] for self-efficacy for easy

tasks, ρICC = 0.84 [0.805,0.88] for medium and ρICC = 0.795 [0.75,0.84] for hard tasks

respectively. In Model 1 we added fixed effects of time and time2.

TITLE: Multivariate multilevel models for change (MVMLMC),
fixed effects (M1) 24 May 2022 ;

DATA: FILE = C:\se.txt ;

VARIABLE: NAMES =
id classid ! student id, classroom id,
mathtop ! content area in maths,
c1_frac6 c2_frac6 c3_frac6 c4_mea6 ! classroom dummy variables
c5_eq10 c6_eq10 c7_eq10 ! classroom dummy variables
c8_geo10 c9_geo10 ! classroom dummy variables
sex grade lesson ! student sex, grade, lesson
seE seM seH ; ! selfefficacy for easy,

! medium and hard tasks ;
USEVAR = seE seM seH time0 time2 ; ! variables included in model
WITHIN = time0 time2 ; ! within-level predictors
CLUSTER = id ; ! clustering variable
MISSING ARE ALL (-9); ! code for missing value
DEFINE:
time0 = lesson - 1 ; ! create time-variable: 0,1,2,3
time2 = time0 * time0 ; ! create time-squared: 0,1,4,9

ANALYSIS: TYPE = TWOLEVEL ; ! two-level model
ESTIMATOR = BAYES ; ! Bayesian estimator
FBITERATIONS = 10000 ; ! number of iterations
PROC = 2 ; ! number of processors
BSEED = 463956 ; ! random seed
THIN = 10 ; ! thinning

MODEL:
%WITHIN% ! within-part of the model
seE ON time0 time2 ; ! self-efficacy (SE) for easy tasks

! regressed on time and time-square
seM ON time0 time2 ; ! SE for medium tasks regressed on time and time-square
seH ON time0 time2 ; ! SE for hard tasks regressed on time and time-square
seE WITH seM seH ; ! associated disturbance terms
seM WITH seH ; ! associated disturbance terms

seE (wv_E) ; ! within level variances with labels
seM (wv_M) ;
seH (wv_H) ;

%BETWEEN% ! between-part of the model
seE WITH seM seH ; ! associated latent intercepts

seM WITH seH ; ! associated latent intercepts
.....

We then, in model 2 (M2) estimated both fixed and random effects of time and

time2, and a sequence of additional parameters for estimating differences in intercepts

(starting-points) and slopes (change) between self-efficacy constructs for the three

difficulty levels.

TITLE: Multivariate multilevel models for change,
fixed and random effects (M2) ;

DATA: FILE = C:\se.txt ;

VARIABLE: NAMES =
id classid ! student id, classroom id,
content ! content area in maths,
c1_frac6 c2_frac6 c3_frac6 c4_mea6 ! classroom dummy variables
c5_eq10 c6_eq10 c7_eq10 ! classroom dummy variables
c8_geo10 c9_geo10 ! classroom dummy variables
sex grade lesson ! student sex, grade, lesson
seE seM seH ; ! selfefficacy for easy,

! medium and hard tasks ;
USEVAR = seE seM seH time0 time2 ; ! variables included in model
WITHIN = time0 time2 ; ! within-level predictors
CLUSTER = id ; ! clustering variable
MISSING ARE ALL (-9); ! code for missing value
DEFINE:
time0 = lesson - 1 ; ! create time-variable 0,1,2,3
time2 = time0 * time0 ; ! create time-squared

..... ! analysis as before

MODEL:
%WITHIN% ! within-part of the model
u1 | seE ON time0 ; ! random term for self-efficacy (SE) for

! easy tasks regressed on time
seE ON time2 ; ! SE for easy tasks regressed on time-square
u2 | seM ON time0 ; ! random term for SE for easy tasks regressed on time
seM ON time2 ; ! SE for medium tasks regressed on time-square
u3 | seH ON time0 ; ! random term for SE for hard tasks regressed on time
seH ON time2 ; ! SE for hard tasks regressed on time-square

seE WITH seM seH ; ! associated disturbance terms
seM WITH seH ; ! associated disturbance terms

%BETWEEN%
[seE] (il); ! name all level-2 latent mean parameters

[seM] (im); ! for comparison (below)
[seH] (ih);
[u1] (vl) ;
[u2] (vm) ;
[u3] (vh) ;

seE (ivl) ; ! name all level-2 latent variance parameters
seM (ivm) ; ! for comparison (below)
seH (ivh) ;
u1 (svl) ;
u2 (svm) ;
u3 (svh) ;

seE WITH seM seH u1 u2 u3 ; ! associations
seM WITH seH u1 u2 u3 ;
seH WITH u1 u2 u3 ;
u1 WITH u2 u3 ;
u2 WITH u3 ;

NEW (int_e_m, int_e_h, int_m_h, ! name new model parameters
slo_e_m, slo_e_h, slo_m_h
intv_e_m, intv_e_h, intv_m_h,
slov_e_m, slov_e_h, slov_m_h) ;

int_e_m = ie - im ; ! difference in intercepts seEASY - seMED
int_e_h = ie - ih ; ! difference in intercepts seEASY - seHARD
int_m_h = im - ih ; ! difference in intercepts seMED - seHARD
slo_e_m = vm - ve ; ! difference in slopes seMED - seEASY
slo_e_h = vh - ve ; ! difference in slopes seHARD - seEASY
slo_m_h = vh - vm ; ! difference in slopes seHARD - seMED

intv_e_m = ivm - ive ; ! difference in intercept var seMED - seEASY
intv_e_h = ivh - ive ; ! difference in intercept var seHARD - seEASY
intv_m_h = ivh - ivm ; ! difference in intercept var seHARD - seMED
slov_e_m = svm - sve ; ! difference in slope var seMED - seEASY
slov_e_h = svh - sve ; ! difference in slope var seHARD - seEASY
slov_m_h = svh - svm ; ! difference in slope var seHARD - seMED
......

We then, in model 3 (M3) included grade (0 = grade 6, 1 = grade 10) as predictor

of the latent intercepts and slopes at the between level, while keeping the within-part of

the model the same as before.

TITLE: Multivariate multilevel models for change,
effect of grade (M3) ;

DATA: FILE = C:\se.txt ;

VARIABLE: NAMES =
id classid ! student id, classroom id,

content ! content area in maths,
c1_frac6 c2_frac6 c3_frac6 c4_mea6 ! classroom dummy variables
c5_eq10 c6_eq10 c7_eq10 ! classroom dummy variables
c8_geo10 c9_geo10 ! classroom dummy variables
sex grade lesson ! student sex, grade, lesson
seE seM seH ; ! selfefficacy for easy,

! medium and hard tasks ;
USEVAR = grade seE seM seH time0 time2 ; ! variables included in model
WITHIN = time0 time2 ; ! within-level predictors
CLUSTER = id ; ! clustering variable
MISSING ARE ALL (-9); ! code for missing value
DEFINE:
time0 = lesson - 1 ; ! create time-variable 0,1,2,3
time2 = time0 * time0 ; ! create time-squared
IF (grade EQ 6) THEN grade_d = 0 ; ! dummy-code grade / year-group
IF (grade EQ 10) THEN grade_d = 1 ;

..... ! analysis as before

MODEL:
%WITHIN% ! within-part of the model as in M2
....

%BETWEEN%
seE WITH seM seH u1 u2 u3 ; ! associated latent variables
seM WITH seH u1 u2 u3 ; ! seE, seM, and seH are latent intercepts
seH WITH u1 u2 u3 ; ! u1, u2, and u3 are latent slopes
u1 WITH u2 u3 ;
u2 WITH u3 ;

seE seM seH u1 u2 u3 ON grade_d ; ! regress six latent variables on grade
.....

We ran Model 4 first among the grade 6 students (M4a), including a

dummy-coded variable for content area of ”measurement” using ”fractions” as baseline.

The content-area dummy-variable was included at the between-level of the model.

TITLE: Multivariate multilevel models for change,
effect of content area in grade 6 (M4a) ;

DATA: FILE = C:\se.txt ;

VARIABLE: NAMES =
id classid ! student id, classroom id,
content ! content area in maths,
c1_frac6 c2_frac6 c3_frac6 c4_mea6 ! classroom dummy variables
c5_eq10 c6_eq10 c7_eq10 ! classroom dummy variables
c8_geo10 c9_geo10 ! classroom dummy variables
sex grade lesson ! student sex, grade, lesson

seE seM seH ; ! selfefficacy for easy,
! medium and hard tasks ;

USEVAR = seE seM seH time0 time2 mea ; ! variables included in model
WITHIN = time0 time2 ; ! within-level predictors
BETWEEN = mea ; ! dummy-variable for "measurement"
CLUSTER = id ; ! clustering variable
MISSING ARE ALL (-9); ! code for missing value
USEOBS (grade EQ 6) ; ! select grade 6 students
DEFINE:
time0 = lesson - 1 ; ! create time-variable 0,1,2,3
time2 = time0 * time0 ; ! create time-squared
IF (c1_frac6 EQ 0) THEN mea = 0 ; ! create dummy code for "measurement" (1)
IF (c2_frac6 EQ 0) THEN mea = 0 ; ! using "fractions" as baseline (0)
IF (c3_frac6 EQ 0) THEN mea = 0 ;
IF (c4_mea6 EQ 0) THEN mea = 0 ;
IF (c1_frac6 EQ 1) THEN mea = 0 ;
IF (c2_frac6 EQ 1) THEN mea = 0 ;
IF (c3_frac6 EQ 1) THEN mea = 0 ;
IF (c4_mea6 EQ 1) THEN mea = 1 ;

..... ! analysis as before

MODEL:
%WITHIN% ! within-part of the model as in M2
....

%BETWEEN%
seE WITH seM seH u1 u2 u3 ; ! associated latent variables
seM WITH seH u1 u2 u3 ; ! seE, seM, and seH are latent intercepts
seH WITH u1 u2 u3 ; ! u1, u2, and u3 are latent slopes
u1 WITH u2 u3 ;
u2 WITH u3 ;

seE seM seH u1 u2 u3 ON mea ;
! regress six latent variables on "measurement" content area

We then ran Model 4 among the grade 10 students (M4b), including a

dummy-variable for content area of ”geometry”, using ”algebra” as baseline. The

content-area dummy-variable was included at the between-level of the model.

TITLE: Multivariate multilevel models for change,
effect of content area in grade 10 (M4b) ;

DATA: FILE = C:\se.txt ;

VARIABLE: NAMES =
id classid ! student id, classroom id,
content ! content area in maths,
c1_frac6 c2_frac6 c3_frac6 c4_mea6 ! classroom dummy variables

c5_eq10 c6_eq10 c7_eq10 ! classroom dummy variables
c8_geo10 c9_geo10 ! classroom dummy variables
sex grade lesson ! student sex, grade, lesson
seE seM seH ; ! selfefficacy for easy,

! medium and hard tasks ;
USEVAR = seE seM seH time0 time2 d_eq ; ! variables included in model
WITHIN = time0 time2 ; ! within-level predictors
BETWEEN = d_eq ; ! dummy-variable for "measurement"
CLUSTER = id ; ! clustering variable
MISSING ARE ALL (-9); ! code for missing value
USEOBS (grade EQ 10) ; ! select grade 10 students
DEFINE:
time0 = lesson - 1 ; ! create time-variable 0,1,2,3
time2 = time0 * time0 ; ! create time-squared
IF (c5_eq10 EQ 0) THEN geo = 0 ; ! create dummy-code for "geometry" (1)
IF (c6_eq10 EQ 0) THEN geo = 0 ; ! using algebra as baseline (0)
IF (c7_eq10 EQ 0) THEN geo = 0 ;
IF (c8_geo10 EQ 0) THEN geo = 0 ;
IF (c9_geo10 EQ 0) THEN geo = 0 ;
IF (c5_eq10 EQ 1) THEN geo = 0 ;
IF (c6_eq10 EQ 1) THEN geo = 0 ;
IF (c7_eq10 EQ 1) THEN geo = 0 ;
IF (c8_geo10 EQ 1) THEN geo = 1 ;
IF (c9_geo10 EQ 1) THEN geo = 1 ;

.......

! in Between model:
seE seM seH u1 u2 u3 ON geo ;
! regress six latent variables on "equation" content area

